"Superlumial" phenomena in linear optics

S. M. Rochester and D. Budker
UC Berkeley Physics Dept.
Sept. 9, 2002

Introduction

Background

Recently, both the scientific and popular (e.g., NYT and NPR) media have paid much atte
optical phenomena dubbed "slow" and "fast" light. When a pulse of weak probe ligh
gates through a nonlinear medium in the presence of a strong drive field, the light pu
under certain circumstances, be transmitted without much distortion, and the *apparen*
velocity can be very slow ($\sim 1\ \text{m s}^{-1}$ in some experiments), faster than c, or negat
recent comprehensive reviews by Matsko et. al., 2001; Boyd and Gauthier, 2002). Although th
studies of these phenomena have been in the context of *nonlinear* optics, in which at l
light fields interact with an atomic medium, these effects can be understood by con
the simpler case of *linear* optics.

In this tutorial, we use considerations from linear optics to show how the variation i
velocity comes about. Although the physics is quite straightforward, and has been
out in detail both theoretically (Garrett and McCumber, 1970) and experimentally (Chu and Wo
we attempt to remove any remaining mystery from the subject by discussing it in ve
language and illustrating each step with graphics derived directly from the discussio
also Sprangle et. al., 2001; McDonald, 2001; Dogariu et. al. 2001).

Approach

We consider an optical medium consisting of an atomic vapor, with light whose frequ
tuned near an isolated resonance. We introduce the complex refractive index, a
describe the propagation of a monochromatic light wave through a slab of the mediu
imaginary part of the refractive index leads to light absorption, while the real part aff

light's phase. Next, we consider a smooth (Gaussian) pulse incident on the slab of our medium. We decompose the pulse into Fourier components, and analyze the propaga each of these components through the medium separately. This is justified in the linear medium whose optical properties are independent of the presence of light. Fina recombine the resulting Fourier components into a "new" pulse, and analyze its spa temporal distribution. Clearly, this discussion does not contain any physical or mathe concepts beyond those found in any introductory physics course. Nevertheless, this sufficient to illustrate all of our conclusions.

Mathematica Setup

Discussion of light propagation in linear media

Monochromatic waves, phase velocity

Definition of the light field

Any plane light wave propagating in the *z* direction can be decomposed into a nur sinusoidal waves. We use the complex notation in which each field is described by an tude times the real part of

```
EField[k_, ω_] = ⅇ^(ⅈ (k z - ω t))
```

$$e^{i(k z - t\omega)}$$

where

```
k[n_, ω_] = n ω / c
```

$$\frac{n\,\omega}{c}$$

is the wavenumber, which depends on the light frequency ω and the index of refractio medium $n(\omega)$. c is the speed of light in vacuum. The real field is given by

```
ReEField[k_, ω_] = ComplexExpand[Re[EField[k, ω]], k]
```

$e^{-z\,\mathrm{Im}[k]}\,\mathrm{Cos}[t\,\omega - z\,\mathrm{Re}[k]]$

```
Table[Plot[ReEField[2 π, 2 π], {z, 0, 8},
   PlotRange → {{0, 8}, {-1.25, 1.25}}], {t, 0, .9, .1}];
```

A monochromatic wave
(double-click plot to view as a function of time)

Phase velocity — general medium

The *phase velocity* is the velocity of a point of constant phase of the electric field, fo ple, a zero crossing. To find the phase velocity we solve for z such that Re[$E(z)$] = function of time:

```
zConstPhase = Solve[ReEField[k, ω] == 0, z]⟦1⟧
```

$\left\{z \to \frac{-\frac{\pi}{2} + t\,\omega}{\mathrm{Re}[k]}\right\}$

The phase velocity is given by $\frac{dz}{dt}$

```
Vph[k_, ω_] = D[z /. zConstPhase, t]
```

$\frac{\omega}{\mathrm{Re}[k]}$

In terms of the index of refraction

```
VphOfN[n_] = ComplexExpand[Vph[k[n, ω], ω], n]
```

$\frac{c}{\mathrm{Re}[n]}$

In a vacuum, n = 1 and the phase velocity is

```
VphOfN[1]
```

```
c
```

Wave propagation through linear atomic media

In a linear medium near an atomic resonance of frequency $\omega 0$ and width Γ, the index o
tion is given by (see, for example, Lipson, 1995, Ch. 13.3.2; Griffiths, 1999, Ch. 8.4)

$$\texttt{nMedium}[\omega\_] = 1 - \frac{\alpha\, c}{\omega - \omega 0 + i\,\Gamma / 2} \frac{\Gamma}{4\,\omega 0}$$

$$1 - \frac{c\,\alpha\,\Gamma}{4\left(\frac{i\,\Gamma}{2} + \omega - \omega 0\right)\omega 0}$$

where α is the absorption coefficient on resonance. The real part of the index of ref
responsible for phase shifts, is given by

$$\texttt{RenMedium}[\omega\_] = \texttt{Collect}\left[\texttt{ComplexExpand}[\texttt{Re}[\texttt{nMedium}[\omega]]] \;/.\; \frac{c\,\alpha\,\Gamma}{\left(\frac{\Gamma^2}{4} + (\omega - \omega 0)^2\right)} \to 4\,\xi,\ \xi\right]$$

$$1 + \xi\left(1 - \frac{\omega}{\omega 0}\right)$$

where we have defined the parameter ξ to be

$$\xi[\omega\_] = \frac{c\,\alpha\,\Gamma}{4\left(\frac{\Gamma^2}{4} + (\omega - \omega 0)^2\right)}$$

$$\frac{c\,\alpha\,\Gamma}{4\left(\frac{\Gamma^2}{4} + (\omega - \omega 0)^2\right)}$$

In a physical medium, the real and the imaginary parts of the refractive index are related through the Kramers - Kron relations (Landau et. al., 1995, Sec. 82). In a simulation, we can easily create an artificial medium that does not pose this property; however, our results immediately become nonphysical, and causality is violated.

The following function plots the real and imaginary parts of the index of refract frequency:

```
IndexPlot[Parameters_, opts___] :=
 Plot[Evaluate[Join[{Re[nMedium[ω]] - 1, Im[nMedium[ω]]},
      If[PlotEnvelope /. {opts} /. PlotEnvelope → False,
       {Γp / 7 Envelope[Δ1 Γ]}, {}]] /. ω → ω0 + Δ1 Γ /. Parameters], {Δ1, -4 , 4},
  PlotStyle → {RGBColor[1, 0, 0], RGBColor[0, 1, 0], RGBColor[0, 0, 1]},
  DisplayFunction → $DisplayFunction,
  FrameLabel → {"Detuning (ω-ω0)/Γ", "Index of refraction"},
  FrameTicks → {True, False, False, False}, PlotLegend → {"Re \!\(\*
StyleBox[\"n\",\nFontSlant->\"Italic\"]\) - 1", "Im \!\(\*
StyleBox[\"n\",\nFontSlant->\"Italic\"]\)",
    StyleForm["Pulse spectrum", FontSize → 12]}, LegendPosition → {.23, .2},
  LegendSize → {.6, .25}, LegendShadow → {.015, -.015}, Axes → True]
```

```
IndexPlot[{c → 1, α → 1, Γ → 2, ω0 → 2 π}];
```

The real and imaginary parts of the index of refraction in a medium as a function of frequency.

The phase velocity in the medium is given by

```
VphMedium[ω_] = Vph[k[RenMedium[ω], ω], ω] // ComplexExpand
```

$$\frac{c}{1+\xi\left(1-\frac{\omega}{\omega 0}\right)}$$

Since ξ is always positive, we see that the phase velocity is c on resonance, less than c resonance, and *greater* than c above resonance. To illustrate this, we can plot the lig tric field as it traverses a medium. The wavenumber in the medium is given by

```
kMedium[ω_] = k[nMedium[ω], ω] // Simplify
```

$$\frac{\omega}{c} - \frac{\alpha\,\Gamma\,\omega}{4\left(\frac{i\,\Gamma}{2} + \omega - \omega 0\right)\omega 0}$$

and the electric field is given by

```
EFieldMedium[z_, t_, ω_] = EField[kMedium[ω], ω] // Simplify
```

$$e^{-\frac{1}{4}\,i\,\omega\left(4\,t - \frac{4\,z}{c} + \frac{z\,\alpha\,\Gamma}{\left(\frac{i\,\Gamma}{2} + \omega - \omega 0\right)\omega 0}\right)}$$

The light intensity is proportional to Abs$[E^2]$, where Abs stands for absolute value. ` verify that the intensity has the correct dependence on the absorption coefficient:

```
Abs[EFieldMedium[z, t, ω0]^2] // ComplexExpand // PowerExpand
```

$$e^{-z\,\alpha}$$

The following function produces an animation of a monochromatic wave trave medium.

```
PlotMonochromatic[{z0_, z1_, z2_, z3_},
  {tMin_, tMax_, tStep_}, Parameters_] := (
  SetOptions[Plot, MaxBend → 6, DisplayFunction → Identity];

  FieldOptions = PlotStyle → {AbsoluteThickness[2]};
  GhostFieldOptions =
   PlotStyle → {AbsoluteThickness[1.5], AbsoluteDashing[{2, 5}]};

  Region1Field[z_, t_] = EFieldMedium[z, t, ω] /. {α → 0} /. Parameters;
  Region2Field[z_, t_] =
   Region1Field[z1, 0] × EFieldMedium[z - z1, t, ω] /. Parameters;
  Region3Field[z_, t_] =
   Region2Field[z2, 0] × EFieldMedium[z - z2, t, ω] /. {α → 0} /. Parameters;

  Table[
   Show[
    Plot[Re[Region1Field[z, t]], {z, z1, z3}, Evaluate[GhostFieldOptions]],
    Plot[Re[Region1Field[z, t]], {z, z0, z1}, Evaluate[FieldOptions]],
    Plot[Re[Region2Field[z, t]], {z, z1, z2}, Evaluate[FieldOptions]],
    Plot[Re[Region3Field[z, t]], {z, z2, z3}, Evaluate[FieldOptions]],
    DisplayFunction → $DisplayFunction, Epilog → {AbsoluteThickness[3],
      Line[{{z1, -1.4}, {z1, 1.4}, {z2, 1.4}, {z2, -1.4}, {z1, -1.4}}]},
    PlotRange → {{z0 - 0.000001, z3}, {-1.5, 1.5}}
   ],
   {t, tMin, tMax - tStep, tStep}];

  SetOptions[Plot, MaxBend → 10, DisplayFunction → $DisplayFunction];
 )
```

Here is a plot of the electric field for a light wave tuned above resonance. The phas wave at the output is advanced relative to what it would have been, had there b medium (dotted line). The amplitude of the wave is reduced because of absorption nary part of the index of refraction). Try changing the parameters and re-evaluating. the light frequency closer to resonance reduces the phase shift and increases absorptio ing below resonance (i.e. making $\omega < \omega 0$), one sees that the phase of the wave at the o retarded relative to what it would have been, had there been no medium (dotted line) l the phase velocity is $< c$ in this case.

```
PlotMonochromatic[{0, 3, 5, 8}, {0., 0.95, .05},
  {c → 1, α → 3., ω → 2.1 π, Γ → 1, ω0 → 2 π}]
```

A monochromatic wave propagating through a linear medium.
(double-click plot to view as a function of time)

Light pulses, group velocity

Constructing a light pulse

Introduction

To construct a Gaussian pulse of width Γp, we superpose waves with amplitudes given

```
Envelope[Δ_] = 1/(Γp Sqrt[2 π]) ⅇ^(-Δ^2/(2 Γp^2))
```

$$\frac{e^{-\frac{\Delta^2}{2\,\Gamma p^2}}}{\sqrt{2\pi}\,\Gamma p}$$

where Δ is the detuning from the central frequency ω, and the first factor provides normalization. The more waves of different frequencies we add, the more the resultar lope looks like a pulse, as we will now show.

Discrete frequencies

Two Frequencies

For illustration, let us look at waves traveling in vacuum, in which $n = 1$. A "pulse" w frequencies is given by

```
TwoFreq =
 Sum[Envelope[Δ] × EField[k[1, ω + Δ], ω + Δ], {Δ, -3 π / 2, 3 π / 2, 3 π}] //
   ComplexExpand // FullSimplify
```

$$\frac{e^{-\frac{9\pi^2}{8\,\Gamma p^2} - i\left(t-\frac{z}{c}\right)\omega} \sqrt{\frac{2}{\pi}}\, \mathrm{Cos}\left[\frac{3\pi\,(c\,t-z)}{2\,c}\right]}{\Gamma p}$$

The envelope is given by the absolute value of the field

```
TwoFreqEnvelope = Abs[TwoFreq] // ComplexExpand // Simplify // PowerExpand
```

$$\frac{e^{-\frac{9\pi^2}{8\,\Gamma p^2}} \sqrt{\frac{2}{\pi}}\, \mathrm{Cos}\left[\frac{3\pi\,(c\,t-z)}{2\,c}\right]}{\Gamma p}$$

```
Re[TwoFreq /. c → 1] // ComplexExpand
```

$$\frac{e^{-\frac{9\pi^2}{8\,\Gamma p^2}} \sqrt{\frac{2}{\pi}}\, \mathrm{Cos}\left[\frac{3}{2}\pi\,(t-z)\right] \times \mathrm{Cos}[(-t+z)\,\omega]}{\Gamma p}$$

Here is the plot. The sinusoidal envelope is an indication of the pulse-like behavior t
come when we add waves of more frequencies.

```
Table[Plot[Evaluate[Re[TwoFreq /. {c → 1, ω → 15 π, Γp → π}]],
   {z, -8 / (2 π), 8 / (2 π)}, PlotRange → {{-8 / (2 π), 8 / (2 π)}, {-.1, .1}},
   PlotPoints → 70], {t, 0, 4 / 3 - 1 / 9, 1 / 9}];
```

Superposition of two monochromatic waves
(double-click plot to view as a function of time)

- **Three frequencies**

Superposing waves of three frequencies, we get

```
ThreeFreq = Sum[Envelope[Δ] × EField[k[1, ω + Δ], ω + Δ], {Δ, -π, π, π}] //
    ComplexExpand // FullSimplify
```

$$\frac{e^{-\frac{\pi^2}{2\,\Gamma p^2}-i\left(t-\frac{z}{c}\right)\omega}\left(e^{\frac{\pi^2}{2\,\Gamma p^2}}+2\,\text{Cos}\left[\pi\left(t-\frac{z}{c}\right)\right]\right)}{\sqrt{2\,\pi}\,\Gamma p}$$

```
Abs[ThreeFreq] // ComplexExpand // Simplify // PowerExpand
```

$$\frac{e^{-\frac{\pi^2}{2\,\Gamma p^2}}\left(e^{\frac{\pi^2}{2\,\Gamma p^2}}+2\,\text{Cos}\left[\pi\left(t-\frac{z}{c}\right)\right]\right)}{\sqrt{2\,\pi}\,\Gamma p}$$

```
Re[ThreeFreq] // ComplexExpand // FullSimplify // PowerExpand
```

$$\frac{e^{-\frac{\pi^2}{2\,\Gamma p^2}}\left(e^{\frac{\pi^2}{2\,\Gamma p^2}}+2\,\text{Cos}\left[\pi\left(t-\frac{z}{c}\right)\right]\right)\text{Cos}\left[\left(-t+\frac{z}{c}\right)\omega\right]}{\sqrt{2\,\pi}\,\Gamma p}$$

```
Table[Plot[Evaluate[Re[ThreeFreq /. {c → 1, ω → 20 π, Γp → π}]], {z,
     -12 / (2 π), 12 / (2 π)}, PlotRange → {{-12 / (2 π), 12 / (2 π)}, {-.3, .3}},
    PlotPoints → 70], {t, 0, 2 - 1 / 9, 1 / 9}];
```

Superposition of three monochromatic waves
(double-click plot to view as a function of time)

- **Four frequencies**

With four frequencies:

```
FourFreq = Sum[Envelope[Δ] × EField[k[1, ω + Δ], ω + Δ], {Δ, -3 π / 2, 3 π / 2, π}]
```

$$\frac{e^{-\frac{9\pi^2}{8\,\Gamma p^2}+i\left(-t\left(-\frac{3\pi}{2}+\omega\right)+\frac{z\left(-\frac{3\pi}{2}+\omega\right)}{c}\right)}}{\sqrt{2\pi}\,\Gamma p}+\frac{e^{-\frac{\pi^2}{8\,\Gamma p^2}+i\left(-t\left(-\frac{\pi}{2}+\omega\right)+\frac{z\left(-\frac{\pi}{2}+\omega\right)}{c}\right)}}{\sqrt{2\pi}\,\Gamma p}+\frac{e^{-\frac{\pi^2}{8\,\Gamma p^2}+i\left(-t\left(\frac{\pi}{2}+\omega\right)+\frac{z\left(\frac{\pi}{2}+\omega\right)}{c}\right)}}{\sqrt{2\pi}\,\Gamma p}+\frac{e^{-\frac{9\pi^2}{8\,\Gamma p^2}+i\left(-t\left(\frac{3\pi}{2}+\omega\right)+\frac{z\left(\frac{3\pi}{2}+\omega\right)}{c}\right)}}{\sqrt{2\pi}\,\Gamma p}$$

```
Table[Plot[Evaluate[Re[FourFreq /. {c → 1, ω → 20 π, Γp → π}]], {z, -12 / (2 π),
    12 / (2 π)}, PlotRange → {{-12 / (2 π), 12 / (2 π)}, {-.5, .5}},
   PlotPoints → 70], {t, 0, 6 / 3 - 1 / 9, 1 / 9}];
```

Superposition of four monochromatic waves
(double-click plot to view as a function of time)

Note that we see periodic "pulse trains" because the frequencies that we use have sma common multiples. We could vary the frequencies to remove this periodic behavior expense of destroying the smooth "pulse-like" appearance. Or, by adding more an frequencies, we can obtain nonperiodic pulses with smooth shape, as shown in tl section.

Many short-pulse lasers operate in the so-called mode-locked regime, emitting a continuous train of pulses separated l a fixed time interval, the round trip time of light in the laser's cavity. The spectrum of such a laser corresponds to "frequency comb," a series of sharp uniformly spaced peaks extending over a spectal range corresponding to the recipr cal of the duration of an individual pulse, so the shorter the pulse, the broader the "comb."

■ Continuous frequency - Gaussian pulse

A real pulse contains every frequency within its bandwidth. Integration over all freq yields the formula

```
GaussianPulse[z_, t_] =
  (Integrate[Envelope[Δ] × EField[k[1, ω0 + Δ], ω0 + Δ], {Δ, -∞, ∞},
    Assumptions → Γp > 0] // FullSimplify)
```

$$e^{-\frac{(c\,t-z)\left((c\,t-z)\,\Gamma p^2+2\,i\,c\,\omega 0\right)}{2\,c^2}}$$

The envelope is, not surprisingly, a Gaussian:

```
Abs[GaussianPulse[z, t]] // ComplexExpand // Simplify // PowerExpand
```

$$e^{-\frac{(-c\,t+z)^2\,\Gamma p^2}{2\,c^2}}$$

```
Table[Plot[Evaluate[Re[GaussianPulse[z, t] /. {c → 1, Γp → π, ω0 → 20 π}]],
    {z, -12 / (2 π), 12 / (2 π)},
    PlotRange → {{-12 / (2 π), 12 / (2 π)}, {-1.5, 1.5}},
    PlotPoints → 70], {t, 0, π, π / 16}];
```

A Gaussian pulse
(double-click plot to view as a function of time)

Group velocity — general transparent medium

Consider two waves with slightly different frequencies

```
Table[ωq = ωc + (-1)^q δ / 2, {q, 1, 2}] // TableForm
```

$$-\frac{\delta}{2} + \omega c$$
$$\frac{\delta}{2} + \omega c$$

Since the difference δ between the freqencies ω_1 and ω_2 is small, we can approxim
frequency dependence of k by a series about the central frequency ωc:

```
kSeries[ω_] = Series[k[ω], {ω, ωc, 1}]
```

$$k[\omega c] + k'[\omega c]\,(\omega - \omega c) + O[\omega - \omega c]^2$$

Here $k'[\omega c] = \frac{d\,k}{d\,\omega}|_{\omega c}$. The wavenumbers of the two waves are given in this approximati

```
Table[k_q = kSeries[ω_q] // Normal, {q, 1, 2}] // TableForm
```

$k[\omega c] - \frac{1}{2}\,\delta\,k'[\omega c]$

$k[\omega c] + \frac{1}{2}\,\delta\,k'[\omega c]$

```
Table[Plot[Evaluate[{Re[EField[k1, ω1]], Re[EField[k2, ω2]]} /.
      {ωc → 6, k[ωc] → 6, k′[ωc] → 1, δ → 1}], {z, 0, 2 π},
   PlotStyle → {RGBColor[0, 0, 0], RGBColor[1, 0, 0]},
   PlotRange → {{0, 2 π}, {-2.5, 2.5}}, PlotPoints → 30],
  {t, 0, 2 π (17 / 20), 2 π / 30}];
```

Two monochromatic waves with different frequencies and wavenumbers
(double-click plot to view as a function of time)

We see that the two waves are "in phase" at some regions of space and time, and are phase in others. Now let us look at the field corresponding to the sum of the two wave by

```
EField[k1, ω1] + EField[k2, ω2]
```

$e^{i\left(-t\left(-\frac{\delta}{2}+\omega c\right)+z\left(k[\omega c]-\frac{1}{2}\delta k'[\omega c]\right)\right)} + e^{i\left(-t\left(\frac{\delta}{2}+\omega c\right)+z\left(k[\omega c]+\frac{1}{2}\delta k'[\omega c]\right)\right)}$

Simplifying:

```
SumField = % // FullSimplify
```

$2\,e^{-i\,(t\,\omega c - z\,k[\omega c])}\,\mathrm{Cos}\left[\frac{1}{2}\,\delta\,(t - z\,k'[\omega c])\right]$

The real field is given by

```
Re[SumField] // ComplexExpand // PowerExpand
```

$2\,\mathrm{Cos}[t\,\omega c - z\,k[\omega c]]\,\mathrm{Cos}\left[\frac{1}{2}\,\delta\,(t - z\,k'[\omega c])\right]$

and the envelope is given by

```
SumFieldEnv = Abs[SumField] // ComplexExpand // PowerExpand
```

$$2\,\mathrm{Cos}\left[\frac{1}{2}\,\delta\,(t - z\,k'[\omega c])\right]$$

We assume here that k is real, i.e. that the medium is transparent.

```
Table[Plot[Evaluate[{Re[SumField], SumFieldEnv, -SumFieldEnv} /.
      {ωc → 6, k[ωc] → 4, k′[ωc] → 1, δ → 1.}], {z, 0, 8 π},
    PlotRange → {{0, 8 π}, {-2.5, 2.5}}, PlotPoints → 50,
    PlotStyle → {RGBColor[0, 0, 0], RGBColor[1, 0, 0], RGBColor[1, 0, 0]}],
  {t, 0, 4 π (19 / 20), 4 π / 20}];
```

Two superposed monochromatic waves in a medium, plotted with their envelope. Note that since the group velocity does not equal the phase velocity, a phase difference between the light field and the envelope is acumulated in time.
(double-click plot to view as a function of time)

To determine the velocity of the pulse, we consider the motion of a given point on th lope, for example, a zero crossing: $z[t]$ such that the $|E[t]| = 0$:

```
zConstantEnv = Solve[SumFieldEnv == 0, z][[1, 1]]
```

$$z \to \frac{-\pi + t\,\delta}{\delta\,k'[\omega c]}$$

The derivative of z with respect to time is the group velocity

```
Vg[k_] = D[z /. zConstantEnv, t]
```

$$\frac{1}{k'[\omega c]}$$

as a function of ω, k is given by

```
kOfω[ω_] = k[n[ω], ω]
```

$$\frac{\omega\, n[\omega]}{c}$$

so the group velocity is given in terms of the index of refraction by

```
VgOfn[n_] = Vg[kOfω] // Simplify
```

$$\frac{c}{n[\omega c] + \omega c\, n'[\omega c]}$$

This derivation is for the case of a transparent medium. As is pointed out in most optic
one has to be careful when applying the concept of group velocity to a medium with
absorption, because pulse reshaping will occur as the pulse travels through the mediu
will look at this more closely next.

Pulse propagation in linear atomic media

■ **Nontransparent media**

We would like to determine when our formula for group velocity in transparent me
apply to nontransparent media. For a Gaussian pulse, the group velocity refers to the v
of the peak of the pulse. Although inside an absorptive medium the pulse shape is d
spatially, we can still measure the group velocity by comparing the times at which th
reaches a maximum at different points inside the medium. Our formula for the group v
should hold as long as there is not too much distortion of the pulse shape, which wou
der the concept of a "group" velocity (the velocity at which the entire pulse travels) m
less. Distortion is caused when resonant spectral components are absorbed much
strongly than off-resonant ones.

Absorption is governed by the imaginary part of k:

```
ImkMedium[ω_] = ⅈ Im[kMedium[ω]] // ComplexExpand // FullSimplify
```

$$\frac{i\,\alpha\,\Gamma^2\,\omega}{2\left(\Gamma^2+4\,(\omega-\omega 0)^2\right)\omega 0}$$

Let's consider a Gaussian input pulse. The spectrum of the input intensity is given by

```
Abs[Envelope[Δ]]^2 // ComplexExpand
```

$$\frac{e^{-\frac{\Delta^2}{\Gamma p^2}}}{2\,\pi\,\Gamma p^2}$$

The spectrum of the output intensity is given by

```
Transmission[Δ_] =
 Abs[Envelope[Δ] × EField[ImkMedium[Δ + ω0], ω] /. Δ + ω0 → ω0]^2 //
   ComplexExpand // PowerExpand
```

$$\frac{e^{-\frac{\Delta^2}{\Gamma p^2}-\frac{z\,\alpha\,\Gamma^2}{\Gamma^2+4\,\Delta^2}}}{2\,\pi\,\Gamma p^2}$$

We assume here that $\omega \simeq \omega 0$.

A Gaussian spectrum has a maxium only in the center. If we require this to be true output spectrum, we can obtain a condition for avoiding pulse reshaping. Setting the tive with respect to Δ equal to zero, we can find the extrema:

```
Extrema = Solve[D[Transmission[Δ], Δ] == 0, Δ]
```

$$\left\{\{\Delta\to 0\},\left\{\Delta\to -\sqrt{-\frac{\Gamma^2}{4}-\frac{1}{2}\sqrt{z}\,\sqrt{\alpha}\,\Gamma\,\Gamma p}\right\},\left\{\Delta\to \sqrt{-\frac{\Gamma^2}{4}-\frac{1}{2}\sqrt{z}\,\sqrt{\alpha}\,\Gamma\,\Gamma p}\right\},\left\{\Delta\to -\sqrt{-\frac{\Gamma^2}{4}+\frac{1}{2}\sqrt{z}\,\sqrt{\alpha}\,\Gamma\,\Gamma p}\right\},\left\{\Delta\to \sqrt{-\frac{\Gamma^2}{4}+\frac{1}{2}\sqrt{z}\,\sqrt{\alpha}\,\Gamma\,\Gamma p}\right\}\right\}$$

The first solution is zero, and the second and third are imaginary. To find the condi

excluding the last two (i.e., requiring them to be imaginary), we solve for $\Delta = 0$ in the
solution:

```
Solve[Δ == 0 /. Extrema[[4]], α]
```

$$\left\{\left\{\alpha \to \frac{\Gamma^2}{4\, z\, \Gamma p^2}\right\}\right\}$$

So we need $\alpha\, z << \Gamma^2 / \Gamma p^2$ to avoid significant pulse reshaping. If this condition is sa
the formula we derived for group velocity in a transparent medium will remain a
mately correct.

```
IndexPlot[{c → 1, α → 1, Γ → 1, ω0 → 2 π, Γp → .1},
  PlotEnvelope → True];
```

The real and imaginary parts of the index of refraction in a medium as a function of frequency, plotted together with the Gaussian pulse spectrum of width Γp.

■ Group velocity

We now apply the expression for group velocity that we derived for transparent me
linear atomic medium near resonance. Substituting our formula for the real part of th
of refraction in a atomic medium into the expression for group velocity, we obtain
group velocity in a linear atomic medium near resonance

```
VgOfn[RenMedium]
```

$$\frac{c}{1 - \frac{\xi\, \omega c}{\omega 0} + \xi \left(1 - \frac{\omega c}{\omega 0}\right)}$$

Using our definition of ξ and assuming that we are directly on resonance gives

```
VgMedium = % /. ξ → ξ[ωc] /. ωc → ω0
```

$$\frac{c}{1 - \frac{c\, \alpha}{\Gamma}}$$

We can see that on resonance, when $\alpha\, c < \Gamma$ the group velocity is greater than c an
$\alpha\, c > \Gamma$ the group velocity is negative. While at first glance, these results may seem u

cal (especially the second, which implies that the pulse moves backwards in the me
they actually have simple, consistent interpretations. We illustrate this below.

It is important to note that there is a change in the spatial width of the pulse upon propagation through an interfa between vacuum and the medium -- the pulse compresses or expands depending on the value of the group velocit Clearly, while the pulse is going through the interface it is severely reshaped; however, we may still discuss tl behaviour of the pulse when most of its spatial extent is contained within the medium or one of the two vacuum regio -- before and after the medium.

Setup for plotting

There is a fast-oscillating phase $e^{-i\,\omega 0\,(t-z)}$ in the electric field that we don't need to if we are plotting the envelope. We'll divide it out of the expression for the E-field:

```
Map[Factor, ExpandAll[EFieldMedium[z, t, ω0 + Δ] ⅇ^(ⅈ ω0 (t-z/c))], {2}]
```

$$e^{-\frac{z\,\alpha\,\Gamma}{2\,(\Gamma-2\,i\,\Delta)}-i\,t\,\Delta+\frac{i\,z\,\Delta}{c}-\frac{z\,\alpha\,\Gamma\,\Delta}{2\,(\Gamma-2\,i\,\Delta)\,\omega 0}}$$

Assuming that $\Gamma << \omega 0$, we can set the term containing $\Gamma / \omega 0$ to zero:

```
EFieldEnv[z_, t_, Δ_] = % /. Γ / ω0 → 0
```

$$e^{-\frac{z\,\alpha\,\Gamma}{2\,(\Gamma-2\,i\,\Delta)}-i\,t\,\Delta+\frac{i\,z\,\Delta}{c}}$$

This function plots a pulse traversing a medium

```
PlotField[FieldFunction_, {z1_, z2_}, EnvelopePlotOptions_] :=
 Plot[Abs[FieldFunction], {z, z1, z2}, Evaluate[Join[EnvelopePlotOptions,
    {DisplayFunction → Identity, PlotDivision → 2, PlotPoints → 10}]]]
```

```
NPlotField[FieldFunction_, {z1_, z2_}, IntRange_, EnvelopePlotOptions_] :=
 Plot[
  Abs[NIntegrate[FieldFunction, {Δ, -IntRange, IntRange}, PrecisionGoal → 3,
    WorkingPrecision → 12, MaxRecursion → 12, MinRecursion → 2]],
  {z, z1, z2}, Evaluate[Join[EnvelopePlotOptions,
    {DisplayFunction → Identity, PlotDivision → 2, PlotPoints → 10}]]]
```

```
PlotPacket[{z0_, z1_, z2_, z3_}, {tMin_, tMax_, tStep_}, Parameters_] := (
  (* Define styles for plotting *)
  FieldOptions = {PlotStyle → {AbsoluteThickness[2]}};
  EnvelopeOptions = {PlotStyle → {AbsoluteThickness[2], RGBColor[1, 0, 0]}};
  GhostFieldOptions =
   {PlotStyle → {AbsoluteThickness[1.5], AbsoluteDashing[{2, 5}]}};
  GhostEnvelopeOptions = {PlotStyle →
     {AbsoluteThickness[1.5], RGBColor[1, 0, 0], AbsoluteDashing[{2, 5}]}};
  (* Define the field for each phase *)
  NoMedField[z_, t_] = GaussianPulse[z, t] /. ω0 → 0 /. Parameters;
  Region1Field[z_, t_, Δ1_] = EFieldEnv[z, t, Δ1] /. {α → 0} /. Parameters;
  Region2Field[z_, t_, Δ1_] =
   Region1Field[z1, 0, Δ1] × EFieldEnv[z - z1, t, Δ1] /. Parameters;
  Region3Field[z_, t_, Δ1_] =
   Region2Field[z2, 0, Δ1] × EFieldEnv[z - z2, t, Δ1] /. {α → 0} /. Parameters;
  R3Mag = Region3Mag /. Parameters /. Region3Mag → 0;
  Env[Δ1_] = Envelope[Δ1] /. Parameters;
  IntRange = 4 Γp /. Parameters;
  (* Draw plots for each region *)
  Table[
   Show[
    PlotField[NoMedField[z, t], {z0, z1}, EnvelopeOptions],
    PlotField[NoMedField[z, t], {z1, z3}, GhostEnvelopeOptions],
    NPlotField[10^(R3Mag (z - z1) / (z2 - z1)) Env[Δ] × Region2Field[z, t, Δ],
     {z1, z2}, IntRange, EnvelopeOptions],
    NPlotField[10^R3Mag Env[Δ] × Region3Field[z, t, Δ],
     {z2, z3}, IntRange, EnvelopeOptions],
    DisplayFunction → $DisplayFunction,
    Axes → {True, False}, Epilog → {AbsoluteThickness[3],
      Line[{{z1, -1.4}, {z1, 1.4}, {z2, 1.4}, {z2, -1.4}, {z1, -1.4}}],
      If[R3Mag ≠ 0, {Text[\!\("×10" \^ R3Mag\), {z2 + .2 (z3 - z2), 1.3}]}, {}]},
    PlotRange → {{z0 - 0.00001, z3}, {0, 1.5}}
   ],
   {t, tMin, tMax, tStep}];
 )
```

Plots of pulse propagation

Superluminal pulse

We start with the case in which $\alpha c < \Gamma$. Here $\alpha c / \Gamma = 0.5$, and the group velocity is

```
VgMedium /. {α → 1 / (2 c), Γ → 1}
```

```
2 c
```

For all the plots below, we set the light frequency to resonance. We see that even tho pulse is partly absorbed (i.e. its amplitude at the output is smaller than that at the inp output pulse has approximately the same shape as the input. The peak of the output advanced relative to where the peak would have been hand there been no medium line), i.e., the pulse appears to travel faster than c inside the medium. No photons trave than c here, rather, the trailing edge of the pulse is absorbed more strongly than the edge creating the illusion of superluminal propagation.

```
PlotPacket[{0., 25., 35., 60.},
 {-30., 90., 3.}, {c → 1, α → .5, Γp → .1, Γ → 1.}]
```

A Gaussian pulse propagating through a linear medium.
(double-click plot to view as a function of time)

- **Negative group velocity**

When $\alpha\, c < \Gamma$, the group velocity is negative:

```
VgMedium /. {α → 6 / c, Γ → 2}
```

$$-\frac{c}{2}$$

This implies that the peak of the output pulse will exit the medium before the peak input pulse enters it:

```
PlotPacket[{0., 25., 35., 60.}, {-30., 90., 3.},
 {c → 1, α → 6., Γp → .1, Γ → 2., Region3Mag → 8}]
```

A Gaussian pulse propagating through a linear medium.
(double-click plot to view as a function of time)

The absorption is now so large that we have to scale the amplitude of the output pu 10^8 in order to see what is going on. Inside the medium the scaling is ramped up ex tially so that the plot remains smooth.

By adjusting the parameters, we can actually see the pulse moving backwards in the m although the 10^{34}scaling required to observe the output makes this situation rather u cal. We can also see that the pulse is compressed inside the medium.

```
VgMedium /. {α → 16 / c, Γ → 4}
```

$$-\frac{c}{3}$$

```
PlotPacket[{0., 25., 35., 60.}, {-30., 90., 3.},
 {c → 1, α → 16., Γp → .1, Γ → 4., Region3Mag → 34}]
```

A Gaussian pulse propagating through a linear medium.
(double-click plot to view as a function of time)

Actually, the spatial pulse-like shape in the medium is an artifact of our scaling. The field at any given time is, in fact, a monotonically decreasing function of *z*. Howeve look at the value of the field at a given point as a function of time, we can observe pu behavior even inside the medium.

Negative group velocities can be observed in experiments in linear optics. A situatio closer to the common experimental setup than the above case is one in which th advance is much smaller than the total pulse width. Here the pulse is much wider tha previous examples, but we still see that the output pulse peaks at the output before th pulse peaks at the input.

```
PlotPacket[{0., 25., 35., 60.}, {-3. × 10^10, 3. × 10^10, 3. × 10^9},
 {c → 1, α → 1., Γp → 10^-10, Γ → 10^-9, Region3Mag → 2}]
```

A Gaussian pulse propagating through a linear medium.
(double-click plot to view as a function of time)

Non-Gaussian pulse

The pulse shape is not restricted to a simple Gaussian, as long as the pulse shape is a the entire pulse spectrum is within the region of linear anomalous dispersic

$\alpha z << \Gamma^2 / \Gamma p^2$, as derived above. Although complex pulse shapes can pass thr medium with group velocity greater than c, information does not travel faster than th of light, since the requirement that the pulse shape is analytic means that all the infor in the pulse is contained in the front "tail" of the pulse. The fact that the peak of th moves faster than c does not mean that the information in the pulse does.

We illustrate the propagation of a non-Gaussian pulse with a double-peaked pulse (t of two Gaussians).

■ Setup for plotting

```
Plot2Packets[{z0_, z1_, z2_, z3_}, {tMin_, tMax_, tStep_}, Parameters_] := (
  (* Define styles for plotting *)
  FieldOptions = {PlotStyle → {AbsoluteThickness[2]}};
  EnvelopeOptions = {PlotStyle → {AbsoluteThickness[2], RGBColor[1, 0, 0]}};
  GhostFieldOptions =
   {PlotStyle → {AbsoluteThickness[1.5], AbsoluteDashing[{2, 5}]}};
  GhostEnvelopeOptions = {PlotStyle →
     {AbsoluteThickness[1.5], RGBColor[1, 0, 0], AbsoluteDashing[{2, 5}]}};
  (* Define the field for each phase *)
  NoMedField[z_, t_] =
   GaussianPulse[z, t] + GaussianPulse[z, t + ts] /. ω0 → 0 /. Parameters;
  Region1Field[z_, t_, Δ1_] =
   EFieldEnv[z, t, Δ1] + EFieldEnv[z, t + ts, Δ1] /. {α → 0} /. Parameters;
  Region2Field[z_, t_, Δ1_] =
   Region1Field[z1, 0, Δ1] × EFieldEnv[z - z1, t, Δ1] /. Parameters;
  Region3Field[z_, t_, Δ1_] =
   Region2Field[z2, 0, Δ1] × EFieldEnv[z - z2, t, Δ1] /. {α → 0} /. Parameters;
  R3Mag = Region3Mag /. Parameters /. Region3Mag → 0;
  Env[Δ1_] = Envelope[Δ1] /. Parameters;
  IntRange = 4 Γp /. Parameters;
  (* Draw plots for each region *)
  Table[
   Show[
    PlotField[NoMedField[z, t], {z0, z1}, EnvelopeOptions],
    PlotField[NoMedField[z, t], {z1, z3}, GhostEnvelopeOptions],
    NPlotField[10^(R3Mag (z - z1) / (z2 - z1)) Env[Δ] × Region2Field[z, t, Δ],
     {z1, z2}, IntRange, EnvelopeOptions],
    NPlotField[10^R3Mag Env[Δ] × Region3Field[z, t, Δ],
     {z2, z3}, IntRange, EnvelopeOptions],
    DisplayFunction → $DisplayFunction,
    Axes → {True, False}, Epilog → {AbsoluteThickness[3],
      Line[{{z1, -1.4}, {z1, 1.4}, {z2, 1.4}, {z2, -1.4}, {z1, -1.4}}],
      If[R3Mag ≠ 0, {Text[\!\("×10" \^ R3Mag\), {z2 + .2 (z3 - z2), 1.3}]}, {}]},
    PlotRange → {{z0 - 0.00001, z3}, {0, 1.5}}
   ],
   {t, tMin, tMax, tStep}];
 )
```

■ Plots

Here we plot a double-peaked pulse propagating through a linear medium. The entir propagates at the same group velocity, and exits the medium advanced relative to v would have been, had there been no medium.

```
Plot2Packets[{0., 25., 35., 60.}, {-30., 90., 3.},
 {c → 1, α → .5, Γp → .1, Γ → 1., ts → 25., Region3Mag → 1}]
```

A double-peaked pulse propagating through a linear medium.
(double-click plot to view as a function of time)

Conclusion

We hope that this tutorial has demonstrated that:

★ All the "unusual" light propagation effects in this system follow directly from the forn complex refractive index and elementary Fourier analysis.

★ Even though it may appear as if the peak of the light pulse is moving faster than *c* certain conditions), nothing is actually traveling that fast. The "superluminal" effec because the back of the pulse is absorbed more than the front, shifting the apparent the pulse forward.

★ The usual formula for the group velocity

$$v_g = \operatorname{Re}\frac{d\omega}{dk} = \frac{c}{\operatorname{Re} n + \omega \operatorname{Re}\frac{dn}{d\omega}},$$

where ω is the light frequency, k is the complex wavenumber, and n is the complex re index, remains valid even in the cases where such velocity becomes infinite or negativ all the seemingly strange consequencies to which this leads.

Acknowledgement and disclaimer

We would like to acknowldge helpful discussions with M. Zolotorev. Creation of this tutorial was s by a UC Berkeley Classroom Technologies Grant (2002). The material is based in part upon work s by the National Science Foundation (NSF) under Grant No. PHY-9733479 and by the Office of Naval (ONR) under Grant No. N00014-97-1-0214. Any opinions, findings, conclusions or recome expressed in this material are those of the authors and do not necessarily reflect the views of the NS ONR.

References

R.W. Boyd and D.J. Gauthier, in *Progress in Optics*, edited by E. Wolf (Elsevier, Amsterdam, 2002), Vol. 43.

R.Y. Chiao, in *Amazing light: a volume dedicated to Charles Hard Townes on his 80th birthday*, edited by R.Y. Chiao (Springer, New York, 1996).

S. Chu and S. Wong, Physical Review Letters **48**, 738 (1982).

A. Dogariu, A. Kuzmich, and L.J. Wang, Physical Review A (Atomic, Molecular, and Optical Physics) **63**, 053806/1 (2001).

C.G.B. Garrett and D.E. McCumber, Physical Review A (General Physics) **1**, 305 (1970).

D. J. Griffiths, *Introduction to electrodynamics* (Prentice Hall, Upper Saddle River, N.J., 1999).

A. Kuzmich, A. Dogariu, L.J. Wang, et al., Physical Review Letters **86**, 3925 (2001).

L.D. Landau, E.M. Lifshits, and L.P. Pitaevskii, *Electrodynamics of continuous media* (Butterworth-Heinemann, Oxford England, 1995).

S.G. Lipson, H. Lipson, and D.S. Tannhauser, *Optical physics* (Cambridge University Press, Cambridge 1995).

A. B. Matsko, O. A. Kocharovskaya, Y. Rostovtsev, et al., in *Advances in Atomic, Molecular, and Optical Physics*, edited by B. Bederson (Academic Press, New York, 2001), Vol. 46, p. 191.

K.T. McDonald, American Journal of Physics **69**, 607 (2001).

P. Sprangle, J.R. Penano, and B. Hafizi, Physical Review E (Statistical, Nonlinear, and Soft Matter Physics) **64**, 026504/1 (2001).